Как получить срез последних если не периодический регистр сведений?
---
примерно так

```1c
    ВЫБРАТЬ
    ИсторияИспользованияШтатногоРасписания.ПозицияШтатногоРасписания КАК ПозицияШтатногоРасписания,
    ИсторияИспользованияШтатногоРасписания.КоличествоСтавок КАК КолСтавКон,
    ИсторияИспользованияШтатногоРасписания.Дата КАК Дата
    ПОМЕСТИТЬ ДанныеРС
    ИЗ
    РегистрСведений.ИсторияИспользованияШтатногоРасписания КАК ИсторияИспользованияШтатногоРасписания
    ГДЕ
    ИсторияИспользованияШтатногоРасписания.Дата <= &КонецПериода
    И ИсторияИспользованияШтатногоРасписания.Используется = ИСТИНА
    И ИсторияИспользованияШтатногоРасписания.Активность = ИСТИНА
    И ИсторияИспользованияШтатногоРасписания.ПозицияШтатногоРасписания = &ПозицияШтатногоРасписания
    ;

    ВЫБРАТЬ
    ДанныеРС.ПозицияШтатногоРасписания КАК ПозицияШтатногоРасписания,
    МАКСИМУМ(ДанныеРС.Дата) КАК Дата
    ПОМЕСТИТЬ ДанныеГруппировки
    ИЗ
    ДанныеРС КАК ДанныеРС
    СГРУППИРОВАТЬ ПО
    ДанныеРС.ПозицияШтатногоРасписания
    ;

    ВЫБРАТЬ
    ДанныеГруппировки.ПозицияШтатногоРасписания КАК ПозицияШтатногоРасписания,
    ДанныеГруппировки.Дата КАК Дата,
    ДанныеРС.КолСтавКон
    ИЗ
    ДанныеГруппировки КАК ДанныеГруппировки
    ВНУТРЕННЕЕ СОЕДИНЕНИЕ ДанныеРС КАК ДанныеРС
    ПО ДанныеГруппировки.ПозицияШтатногоРасписания = ДанныеРС.ПозицияШтатногоРасписания
    И ДанныеГруппировки.Дата = ДанныеРС.Дата
```
---
Ну это надо сначала определить, что такое "срез последних если не периодический регистр сведений". Трактовать можно по-разному о)
---
Смешно, да.
---
По определению у непериодического РС нет "срез последних".
Надо сделать сперва РС периодическим.
---
Гугли "срез последних на каждую дату в запросе".
Там методика "ручного" среза последних
---
Нужно последнюю запись на дату
---
Максимум по дате
---
Вот так:
```1c
    ВЫБРАТЬ
    ИсторияИспользованияШтатногоРасписания.ПозицияШтатногоРасписания КАК ПозицияШтатногоРасписания,
    СУММА(ИсторияИспользованияШтатногоРасписания.КоличествоСтавок) КАК КолСтавКон,
    МАКСИМУМ(ИсторияИспользованияШтатногоРасписания.Дата) КАК Дата
    ИЗ
    РегистрСведений.ИсторияИспользованияШтатногоРасписания КАК ИсторияИспользованияШтатногоРасписания
    ГДЕ
    ИсторияИспользованияШтатногоРасписания.Дата <= &КонецПериода
    И ИсторияИспользованияШтатногоРасписания.Используется = ИСТИНА
    И ИсторияИспользованияШтатногоРасписания.Активность = ИСТИНА
    И ИсторияИспользованияШтатногоРасписания.ПозицияШтатногоРасписания = &ПозицияШтатногоРасписания

    СГРУППИРОВАТЬ ПО
    ИсторияИспользованияШтатногоРасписания.ПозицияШтатногоРасписания,
    ИсторияИспользованияШтатногоРасписания.Дата

    УПОРЯДОЧИТЬ ПО
    ПозицияШтатногоРасписания,
    Дата
```
---
Но так не получается
---
Наверно сначала нужно по нужным позициям найти максимальную дату, а потом на эту дату получить нужную позицию...
---
В одном запросе делаешь максимум по дате для ПозицияШтатногоРасписания и помещаешь во временную таблицу. Вторым запросом соединяешься по этой дате и ПозицияШтатногоРасписания с РС и получаешь из него количество ставок.
---
А можно пример, пожалуйста
---
примерно так

```1c
    ВЫБРАТЬ
    ИсторияИспользованияШтатногоРасписания.ПозицияШтатногоРасписания КАК ПозицияШтатногоРасписания,
    ИсторияИспользованияШтатногоРасписания.КоличествоСтавок КАК КолСтавКон,
    ИсторияИспользованияШтатногоРасписания.Дата КАК Дата
    ПОМЕСТИТЬ ДанныеРС
    ИЗ
    РегистрСведений.ИсторияИспользованияШтатногоРасписания КАК ИсторияИспользованияШтатногоРасписания
    ГДЕ
    ИсторияИспользованияШтатногоРасписания.Дата <= &КонецПериода
    И ИсторияИспользованияШтатногоРасписания.Используется = ИСТИНА
    И ИсторияИспользованияШтатногоРасписания.Активность = ИСТИНА
    И ИсторияИспользованияШтатногоРасписания.ПозицияШтатногоРасписания = &ПозицияШтатногоРасписания
    ;

    ВЫБРАТЬ
    ДанныеРС.ПозицияШтатногоРасписания КАК ПозицияШтатногоРасписания,
    МАКСИМУМ(ДанныеРС.Дата) КАК Дата
    ПОМЕСТИТЬ ДанныеГруппировки
    ИЗ
    ДанныеРС КАК ДанныеРС
    СГРУППИРОВАТЬ ПО
    ДанныеРС.ПозицияШтатногоРасписания
    ;

    ВЫБРАТЬ
    ДанныеГруппировки.ПозицияШтатногоРасписания КАК ПозицияШтатногоРасписания,
    ДанныеГруппировки.Дата КАК Дата,
    ДанныеРС.КолСтавКон
    ИЗ
    ДанныеГруппировки КАК ДанныеГруппировки
    ВНУТРЕННЕЕ СОЕДИНЕНИЕ ДанныеРС КАК ДанныеРС
    ПО ДанныеГруппировки.ПозицияШтатногоРасписания = ДанныеРС.ПозицияШтатногоРасписания
    И ДанныеГруппировки.Дата = ДанныеРС.Дата
```
---
вопрос на засыпку!

а если у вас на последнюю, максимальную (одинаковую) дату несколько записей по измерению?
(в непериодическом такое может быть)
то какую из 10 записей, с разными ресурсами надо брать ?
---
Спасибо большое)